Escaping Down a Rope

In[1]:=
```
Clear["Global`*"]
```

In[2]:=
```
y0 = 15; v0 = 0; g = 9.8; W = 600; m = Abs[W / g]
L = 20 (* Length of rope*);
T = 360 (* max tension for rope *);
```

Out[2]= 61.2245

Take the 20m rope. At the top of the window, start with a doubled loop of length ‘d’. That is strong enough to hold the student (Tmax = 720N for a doubled rope) so the student can slide to the end of that doubled section with final velocity v1 = 0.

In[5]:=
```
y1[d_] := y0 - d;
v1[d_] := 0;
(* Slide to bottom of doubled section and start next stage from rest. *)
```

Next, slide down the remaining section of the rope till you either reach the ground or run out of rope. The remaining rope is L - 2 d. y2 will be the height at the end of that slide.

In[7]:=
```
y2[d_] := Max[y1[d] - (L - 2 d), 0]
```

In[8]:=
```
a = (T - W) / m
```

Out[8]= -3.92

In[9]:=
```
v2[d_] := -Sqrt[v1[d]^2 + 2 a (y2[d] - y1[d])]
```

Finally, freefall from y2 down to ground. (If y2 is already 0, this does nothing.)

In[10]:=
```
y3 = 0;
v3[d_] := -Sqrt[(v2[d])^2 + 2 (-g) (y3 - y2[d])]
```

In[12]:=
```
v3[d]
```

Out[12]= $-\sqrt{7.84\ (15 - d - \text{Max}[0,\ -5 + d]) + 19.6\ \text{Max}[0,\ -5 + d]}$

In[13]:=
```
TableForm[
 Table[{d, y1[d], v1[d], y2[d], v2[d], y3, v3[d]}, {d, 0, 10}],
 TableHeadings → {None, {"d", "y1", "v1", "y2", "v2", "final y", "final v"}}]
```

Out[13]//TableForm=

d	y1	v1	y2	v2	final y	final v
0	15	0	0	-10.8444	0	-10.8444
1	14	0	0	-10.4766	0	-10.4766
2	13	0	0	-10.0955	0	-10.0955
3	12	0	0	-9.69948	0	-9.69948
4	11	0	0	-9.28655	0	-9.28655
5	10	0	0	-8.85438	0	-8.85438
6	9	0	1	-7.9196	0	-9.07304
7	8	0	2	-6.85857	0	-9.28655
8	7	0	3	-5.6	0	-9.49526
9	6	0	4	-3.9598	0	-9.69948
10	5	0	5	0.	0	-9.89949

In[14]:=
```
Plot[-v3[d], {d, 0, 10}, LabelStyle → Larger, Frame → True,
 FrameLabel → {"d", "Impact speed"}, ImageSize → Large]
```

Out[14]=



In[15]:=
```
Minimize[{-v3[d], 0 ≤ d ≤ 10}, d]
```

Out[15]= {8.85438, {d → 5.}}

The optimal strategy is to make a loop of 5m at the top (which uses 10 m of rope). This leaves you 10m off the ground. Sliding down the remaining 10m of rope with a = -3.92 m/s^2 has you reach the bottom with a speed of 8.85 m/s.